>>>+
[
	[-]
	>>
	[-]
	++>+>+++++++
	[<++++>>++<-]
	++>>+>+>+++++
	[>++>++++++<<-]
	+>>>,<++
	[
		[
			>
			[->>]
			<
			[>>]
			<<-
		]
		<
		[<]
		<+>>
		[>]
		>
		[
			<+>-
			[
				[<+>-]
				>
			]
			<
			[
				[
					[-]
					<
				]
				++<-
				[
					<+++++++++>
					[<->-]
					>>
				]
				>>
			]
		]
		<<
	]
	<
]
<
[
	[<]
	>
	[
		[>]
		>>
		[>>]
		+
		[<<]
		<
		[<]
		<+>>-
	]
	>
	[>]
	+
	[->>]
	<<<<
	[
		[<<]
		<
		[<]
		+<<
		[
			+>+<<-
			[
				>-->+<<-
				[
					>+<
					[>>+<<-]
				]
			]
			>
			[<+>-]
			<
		]
		++>>-->
		[>]
		>>
		[>>]
	]
	<<
	[
		>>+<
		[
			[<]
			<
		]
		>
		[
			[<<]
			<
			[<]
			+
			[
				-<+>>-
				[
					<<+>++>-
					[
						<->
						[<<+>>-]
					]
				]
				<
				[>+<-]
				>
			]
			>
			[>]
			>
		]
		>
		[>>]
		>>
	]
	<<
	[>>+>>+>>]
	<<
	[->>>>>>>>]
	<<
	[>.>>>>>>>]
	<<
	[>->>>>>]
	<<
	[>,>>>]
	<<
	[>+>]
	<<
	[+<<]
	<
]